```
> restart:
```

Uniform rods of mass m and length l, center of mass positions:

```
> x[1] :=  l/2 * sin(theta[1](t));
> y[1] := -l/2 * cos(theta[1](t));
> x[2] :=  l/2 * sin(theta[2](t)) + 2*x[1];
> y[2] := -l/2 * cos(theta[2](t)) + 2*y[1];
```

$$x_1 := \frac{l \sin(\theta_1(t))}{2}$$

$$y_1 := -\frac{l \cos(\theta_1(t))}{2}$$

$$x_2 := \frac{l \sin(\theta_2(t))}{2} + l \sin(\theta_1(t))$$

$$y_2 := -\frac{l \cos(\theta_2(t))}{2} - l \cos(\theta_1(t)) \tag{1}$$

Pendulum's end position:

```
> x[e] :=  l * sin(theta[2](t)) + 2*x[1];
> y[e] := -l * cos(theta[2](t)) + 2*y[1];
```

$$x_e := l \sin(\theta_2(t)) + l \sin(\theta_1(t))$$

$$y_e := -l \cos(\theta_2(t)) - l \cos(\theta_1(t)) \tag{2}$$

Kinetic energy, center of mass linear motion:

```
> K := m/2 * simplify(diff(x[1],t)^2 + diff(y[1],t)^2 + diff(x[2],t)
  ^2 + diff(y[2],t)^2);
```

$$K := \frac{1}{8}\left(m\left(5\left(\frac{d}{dt}\theta_1(t)\right)^2 + 4\left(\frac{d}{dt}\theta_2(t)\right)\left(\cos(\theta_2(t))\cos(\theta_1(t)) + \sin(\theta_2(t))\sin(\theta_1(t))\right)\left(\frac{d}{dt}\theta_1(t)\right) + \left(\frac{d}{dt}\theta_2(t)\right)^2\right)l^2\right) \tag{3}$$

Kinetic energy, rod rotation (moment of intertia of uniform rod around center of mass is m*l^2/12):

```
> J := 1/2 * m*l^2/12 * (diff(theta[1](t),t)^2 + diff(theta[2](t),t)
  ^2);
```

$$J := \frac{m\, l^2\left(\left(\frac{d}{dt}\theta_1(t)\right)^2 + \left(\frac{d}{dt}\theta_2(t)\right)^2\right)}{24} \tag{4}$$

Pendulum's end is attached to a point (X,Y)*l by a spring of stiffness k:

```
> V := k/2 * ( (x[e] - l*X)^2 + (y[e] - l*Y)^2 );
```

$$V := \frac{1}{2}\left(k\left(\left(-l X + l\sin\left(\theta_2(t)\right) + l\sin\left(\theta_1(t)\right)\right)^2 + \left(-l Y - l\cos\left(\theta_2(t)\right) - l\cos\left(\theta_1(t)\right)\right)^2\right)\right) \tag{5}$$

Proper Lagrangian (in units of m*l^2/2):

```
> subs(k=m/2*omega^2, (K + J - V)): expand(%/(m*l^2/2)): combine(%):
> subs(['diff(theta[i](t),t) = v[i]' $ i = 1..2], %):
> subs(['theta[i](t) = theta[i]' $ i = 1..2], %): L := %;
```

$$L := v_2 v_1 \cos\left(-\theta_2 + \theta_1\right) + \frac{4 v_1^2}{3} + \frac{v_2^2}{3} - \frac{\omega^2 X^2}{2} + \omega^2 X \sin\left(\theta_2\right) + \omega^2 X \sin\left(\theta_1\right) - \omega^2 - \omega^2 \cos\left(-\theta_2 + \theta_1\right) - \frac{\omega^2 Y^2}{2} - \omega^2 Y \cos\left(\theta_2\right) - \omega^2 Y \cos\left(\theta_1\right) \tag{6}$$

Canonical momenta:

```
> p[1] := diff(L, v[1]); p[2] := diff(L, v[2]);
```

$$p_1 := v_2 \cos\left(-\theta_2 + \theta_1\right) + \frac{8 v_1}{3}$$

$$p_2 := v_1 \cos\left(-\theta_2 + \theta_1\right) + \frac{2 v_2}{3} \tag{7}$$

Total energy:

```
> E := simplify(v[1]*p[1] + v[2]*p[2] - L);
```

$$E := \frac{\left(6\omega^2 + 6 v_2 v_1\right)\cos\left(-\theta_2 + \theta_1\right)}{6} + \omega^2 Y \cos\left(\theta_1\right) + \omega^2 Y \cos\left(\theta_2\right) - \omega^2 X \sin\left(\theta_1\right) - \omega^2 X \sin\left(\theta_2\right) + \frac{\left(3 X^2 + 3 Y^2 + 6\right)\omega^2}{6} + \frac{4 v_1^2}{3} + \frac{v_2^2}{3} \tag{8}$$

Angular velocities in terms of generalized momenta:

```
> solve([p[1]=pi[1], p[2]=pi[2]], [v[1], v[2]]): op(%);
```

$$\left[v_1 = \frac{3\left(3\pi_2 \cos\left(-\theta_2 + \theta_1\right) - 2\pi_1\right)}{9\cos\left(-\theta_2 + \theta_1\right)^2 - 16},\ v_2 = \frac{3\left(3\cos\left(-\theta_2 + \theta_1\right)\pi_1 - 8\pi_2\right)}{9\cos\left(-\theta_2 + \theta_1\right)^2 - 16}\right] \tag{9}$$

Euler-Lagrange equations:

```
> dp[1]/dt = diff(L, theta[1]); dp[2]/dt = diff(L, theta[2]);
```

$$\frac{dp_1}{dt} = -v_2 v_1 \sin\left(-\theta_2 + \theta_1\right) + \omega^2 X \cos\left(\theta_1\right) + \omega^2 \sin\left(-\theta_2 + \theta_1\right) + \omega^2 Y \sin\left(\theta_1\right)$$

(10)

$$\frac{dp_2}{dt} = v_2\, v_1 \sin\left(-\theta_2 + \theta_1\right) + \omega^2\, X \cos\left(\theta_2\right) - \omega^2 \sin\left(-\theta_2 + \theta_1\right) + \omega^2\, Y \sin\left(\theta_2\right) \tag{10}$$