```
> restart:
```

Uniform rods of mass m and length l, center of mass positions:

```
> x[1] :=  l/2 * sin(theta[1](t));
> y[1] := -l/2 * cos(theta[1](t));
> x[2] :=  l/2 * sin(theta[2](t)) + 2*x[1];
> y[2] := -l/2 * cos(theta[2](t)) + 2*y[1];
```

$$x_1 := \frac{l \sin(\theta_1(t))}{2}$$

$$y_1 := -\frac{l \cos(\theta_1(t))}{2}$$

$$x_2 := \frac{l \sin(\theta_2(t))}{2} + l \sin(\theta_1(t))$$

$$y_2 := -\frac{l \cos(\theta_2(t))}{2} - l \cos(\theta_1(t)) \tag{1}$$

Pendulum's middle position and velocity:

```
> x[m] :=  2*x[1]; y[m] := 2*y[1];
> u[x] := diff(x[m], t); u[y] := diff(y[m], t);
```

$$x_m := l \sin(\theta_1(t))$$

$$y_m := -l \cos(\theta_1(t))$$

$$u_x := l \left(\frac{d}{dt} \theta_1(t)\right) \cos(\theta_1(t))$$

$$u_y := l \left(\frac{d}{dt} \theta_1(t)\right) \sin(\theta_1(t)) \tag{2}$$

Pendulum's end position and velocity:

```
> x[e] :=  l * sin(theta[2](t)) + 2*x[1];
> y[e] := -l * cos(theta[2](t)) + 2*y[1];
> v[x] := diff(x[e], t); v[y] := diff(y[e], t);
```

$$x_e := l \sin(\theta_2(t)) + l \sin(\theta_1(t))$$

$$y_e := -l \cos(\theta_2(t)) - l \cos(\theta_1(t))$$

$$v_x := l \left(\frac{d}{dt} \theta_2(t)\right) \cos(\theta_2(t)) + l \left(\frac{d}{dt} \theta_1(t)\right) \cos(\theta_1(t))$$

$$v_y := l \left(\frac{d}{dt} \theta_2(t)\right) \sin(\theta_2(t)) + l \left(\frac{d}{dt} \theta_1(t)\right) \sin(\theta_1(t)) \tag{3}$$

Kinetic energy, center of mass linear motion:

```
> K := m/2 * simplify(diff(x[1],t)^2 + diff(y[1],t)^2 + diff(x[2],t)
```

```
  ^2 + diff(y[2],t)^2);
```

$$K := \frac{1}{8}\left(m\left(5\left(\frac{d}{dt}\theta_1(t)\right)^2 + 4\left(\frac{d}{dt}\theta_2(t)\right)\left(\cos(\theta_2(t))\cos(\theta_1(t)) + \sin(\theta_2(t))\sin(\theta_1(t))\right)\left(\frac{d}{dt}\theta_1(t)\right) + \left(\frac{d}{dt}\theta_2(t)\right)^2\right) l^2\right) \quad (4)$$

*Kinetic energy, rod rotation (moment of intertia of uniform rod around center of mass is m*l^2/12):*

```
> J := 1/2 * m*l^2/12 * (diff(theta[1](t),t)^2 + diff(theta[2](t),t)
  ^2);
```

$$J := \frac{m l^2 \left(\left(\frac{d}{dt}\theta_1(t)\right)^2 + \left(\frac{d}{dt}\theta_2(t)\right)^2\right)}{24} \quad (5)$$

Potential energy of the rod's center of masses:

```
> V := m*g*(y[1] + y[2]);
```

$$V := m g \left(-\frac{3 l \cos(\theta_1(t))}{2} - \frac{l \cos(\theta_2(t))}{2}\right) \quad (6)$$

*Proper Lagrangian (in units of m*l^2/2):*

```
> subs(g=l*omega^2, (K + J - V)): expand(%/(m*l^2/2)): combine(%):
> subs(['diff(theta[i](t),t) = v[i]' $ i = 1..2], %):
> subs(['theta[i](t) = theta[i]' $ i = 1..2], %): L := %;
```

$$L := v_2 v_1 \cos(-\theta_2 + \theta_1) + \frac{4 v_1^2}{3} + \frac{v_2^2}{3} + 3\omega^2 \cos(\theta_1) + \omega^2 \cos(\theta_2) \quad (7)$$

Canonical momenta:

```
> p[1] := diff(L, v[1]); p[2] := diff(L, v[2]);
```

$$p_1 := v_2 \cos(-\theta_2 + \theta_1) + \frac{8 v_1}{3}$$

$$p_2 := v_1 \cos(-\theta_2 + \theta_1) + \frac{2 v_2}{3} \quad (8)$$

Total energy:

```
> E := simplify(v[1]*p[1] + v[2]*p[2] - L);
```

$$E := v_2 v_1 \cos(-\theta_2 + \theta_1) - 3\omega^2 \cos(\theta_1) - \omega^2 \cos(\theta_2) + \frac{4 v_1^2}{3} + \frac{v_2^2}{3} \quad (9)$$

Angular velocities in terms of generalized momenta:

```
> solve([p[1]=pi[1], p[2]=pi[2]], [v[1], v[2]]): op(%);
```

(10)

$$\left[v_1 = \frac{3\left(3\,\pi_2\cos\left(-\theta_2+\theta_1\right)-2\,\pi_1\right)}{9\cos\left(-\theta_2+\theta_1\right)^2-16},\ v_2 = \frac{3\left(3\cos\left(-\theta_2+\theta_1\right)\pi_1-8\,\pi_2\right)}{9\cos\left(-\theta_2+\theta_1\right)^2-16}\right] \quad \textbf{(10)}$$

Euler-Lagrange equations:

```
> dp[1]/dt = diff(L, theta[1]); dp[2]/dt = diff(L, theta[2]);
```

$$\frac{dp_1}{dt} = -v_2\,v_1\sin\left(-\theta_2+\theta_1\right)-3\,\omega^2\sin\left(\theta_1\right)$$

$$\frac{dp_2}{dt} = v_2\,v_1\sin\left(-\theta_2+\theta_1\right)-\omega^2\sin\left(\theta_2\right) \quad \textbf{(11)}$$

Rayleigh dissipation function:

```
> u[x]^2 + u[y]^2 + v[x]^2 + v[y]^2:
> subs(['diff(theta[i](t),t) = v[i]' $ i = 1..2], %):
> subs(['theta[i](t) = theta[i]' $ i = 1..2], %): F := gamma/l^2/2 *
  %;
```

$$F := \frac{1}{2\,l^2}\left(\gamma\left(\left(l^2\,v_1^2\cos\left(\theta_1\right)^2+l^2\,v_1^2\sin\left(\theta_1\right)^2+\left(l\,v_2\cos\left(\theta_2\right)+l\,v_1\cos\left(\theta_1\right)\right)^2\right.\right.\right.$$
$$\left.\left.\left.+\left(l\,v_2\sin\left(\theta_2\right)+l\,v_1\sin\left(\theta_1\right)\right)^2\right)\right) \quad \textbf{(12)}$$

Generalized dissipation forces:

```
> f[1] := combine(diff(F, v[1])); f[2] := combine(diff(F, v[2]));
```

$$f_1 := \gamma\cos\left(-\theta_2+\theta_1\right)v_2+2\,\gamma\,v_1$$

$$f_2 := \gamma\cos\left(-\theta_2+\theta_1\right)v_1+v_2\,\gamma \quad \textbf{(13)}$$